

## Il tipo dati Variant

Tutte le variabili, a cui non è stato assegnato uno specifico tipo di dato, sono di tipo Variant. Nella pratica, se non vogliamo restare relegati ad uno specifico tipo di dati ci conviene usare dati Variant.

Questo non viene fatto di sovente per non occupare troppa memoria ma ci si limita ai casi in cui risulta indispensabile farlo.

Osserviamo, infine, che il contenuto di una cella in Excel è di tipo Variant, difatti può contenere sia dati di tipo numerico o tipo alfanumerico (*stringhe*).

## Funzione di tipo Variant

All'interno di Excel sono predefinite delle funzioni in grado di restituire, oltre al singolo valore, più valori. Ad esempio, la funzione MATR.INVERSA (X) che restituisce una matrice di valori (la matrice inversa di X).

In questa lezione vedremo come si può costruire una funzione definita dall'utente in grado di far questo. Lo faremo attraverso un esempio: le soluzioni di un'equazione algebrica di secondo grado.

Sappiamo, dallo studio dell'algebra, che un'equazione algebrica di secondo grado, nel campo dei numeri complessi, ammette sempre due radici che saranno:

- reali e distinte (se il discriminante  $\Delta > 0$ );
- reali e coincidenti (se il discriminante  $\Delta = 0$ );
- complesse e coniugate (se il discriminante  $\Delta < 0$ ).

## Equazione di secondo grado in VBA

Definiamo in VBA la seguente funzione:

```
Public Function Eq(a As Double, b As Double, c As Double) As Variant
Dim d As Double
Dim v() As Variant
ReDim v(2)           'Ridimensiona la variabile
If a = 0 Then
    v(0) = "Errore"   'Equazione non di 2° grado
    v(1) = ""
    v(2) = ""
Else
    d = b ^ 2 - 4 * a * c
    Select Case d
        Case Is >= 0
            v(0) = "Sol. reali"
            v(1) = (-b - Sqr(d)) / (2 * a)
            v(2) = (-b + Sqr(d)) / (2 * a)
        Case Is < 0
            v(0) = "Sol. complesse"
            v(1) = (-b) / (2 * a) & " + " & (Sqr(-d)) / (2 * a) & " i"
```

```

v(2) = (-b) / (2 * a) & " - " & (Sqr(-d)) / (2 * a) & " i"
End Select
End If
Eq = v
End Function

```

Le soluzioni vengono poste nel vettore v di tipo Variant, dove:

- v(0) contiene un messaggio esplicativo sul tipo di soluzione;
- v(1) contiene la prima soluzione dell'equazione;
- v(2) contiene la seconda soluzione dell'equazione.

Alla fine si assegna alla funzione Eq il valore v.

Si è preferito usare l'istruzione **SELECT CASE**, anziché un **IF** nidificato, per rendere più facilmente leggibile e auto-esplicativo il listato.

Si noti, infine, che l'operatore & è un concatenatore di stringhe.

## Utilizzo della funzione Eq all'interno di Excel

Visualizziamo l'organizzazione del foglio attraverso alcuni esempi che testano il corretto funzionamento della funzione Eq.

Caso in cui  $\Delta > 0$ .

	A	B	C	D	E	F	G	H	I	
1	<b>Equazione di secondo grado</b>									
2	Coefficienti							Soluzione		
3		a	b	c			Nota	x1	x2	
4		1	-5	6			Sol. reali	2	3	
5										

Caso in cui  $\Delta < 0$ .

	A	B	C	D	E	F	G	H	I	
1	<b>Equazione di secondo grado</b>									
2	Coefficienti							Soluzione		
3		a	b	c			Nota	x1	x2	
4		2	0	18			Sol. complesse	0 + 3 i	0 - 3 i	
5										

Caso in cui  $\Delta = 0$ .

	A	B	C	D	E	F	G	H	I	
1	<b>Equazione di secondo grado</b>									
2	Coefficienti							Soluzione		
3		a	b	c			Nota	x1	x2	
4		4	-12	9			Sol. reali	1,5	1,5	
5										

Caso in cui  $a = 0$ . Non è un'equazione di secondo grado.

	A	B	C	D	E	F	G	H	I
1	<b>Equazione di secondo grado</b>								
2	Coefficienti					Soluzione			
3	a	b	c				Nota	x1	x2
4	0	1	2				Errore		
5									
6									

Mostriamo ora il piano delle formule

	A	B	C	D	E	F	G	H	I
1	<b>Equazione di secondo grado</b>								
2	Coefficienti					Soluzione			
3	a	b	c				Nota	x1	x2
4	0	1	2				=Eq(B4;C4;D4)	=Eq(B4;C4;D4)	=Eq(B4;C4;D4)
5									
6									

L'equazione viene definita nella cella [G4], poi si seleziona l'intervallo [G4;I4], si preme il tasto **F2** (modifica formula) quindi si premono i tasti **CTRL + SHIFT + INVIO** in modo che la formula venga estesa a tutto l'intervallo.

*Prof. Ettore Limoli*